



DECUS

PROGRAM LIBRARY

DECUS NO.	8-515
TITLE	PROGRAM TO MATE PAL III WITH SYMBOLIC EDITOR
AUTHOR	G. Chase
COMPANY	Portsmouth Abbey School Portsmouth, Rhode Island
DATE	March 13, 1972
SOURCE LANGUAGE	PAL III

ATTENTION

This is a USER program. Other than requiring that it conform to submittal and review standards, no quality control has been imposed upon this program by DECUS.

The DECUS Program Library is a clearing house only; it does not generate or test programs. No warranty, express or implied, is made by the contributor, Digital Equipment Computer Users Society or Digital Equipment Corporation as to the accuracy or functioning of the program or related material, and no responsibility is assumed by these parties in connection therewith.

DECUS

PROGRAM LIBRARY



1960-1961

1962-1963

1964-1965

1966-1967

1968-1969

1970-1971

1972-1973

1974-1975

1976-1977

1978-1979

1980-1981

1982-1983

1984-1985

PROGRAM TO MATE PAL-III WITH SYMBOLIC EDITOR

Hardware: 8-K PDP-8 series computer with Teletype or equivalent

Software: PAL-III Assembler
Symbolic Editor

Core Required: Page 37 (octal) in Pal's core field (field 0 or 1, at the user's option). The binary loader must reside elsewhere if it is to load Mate.

Function: Enables PAL-III to read Symbolic Editor's text buffer directly, without paper tape input.

Restrictions: 1.) Will operate between core fields 0 and 1 only.
2.) The low-speed reader input routine in PAL-III is overlaid in two places (loc. 1401 and 1403). All other options remain.
3.) Text limitations are those of Symbolic Editor, which Mate does not overlay or modify. Cf. DEC manuals for details.

Additional Features: Two methods of paginating the third-pass listing from PAL-III are provided.

Use: 1.) Load PAL-III into field 0 or field 1, using a binary loader not resident in that field.
2.) Load Mate into Pal's field.
3.) Load Editor into field 1 or field 0--whichever is free.
4.) Start Editor in its field in the usual way. Read ("R") or type ("A") your source program, using tab codes to provide format. Edit as desired. Halt the computer.
(No paper tape need be punched.)
5.) Set both D.F. and I.F. to Pal's field. Start Pal in its field as usual. All normal user options remain; the only change is the lack of paper tape input.

Pagination:
(3rd pass) 1.) Mate provides an automatic page feed (FORM + return = 13 line feeds) every 55 lines of text.
2.) The user can pre-empt the automatic page feed by typing CTRL/K + Return in his source program. Mate promotes CTRL/K to CTRL/L (= FORM), passes it on to Pal, and zeroes the line counter.
3.) Location 7624 contains the negative of the line count per page. Normally this is 7711 = -67 octal = -55 decimal. The user may change this to another value. Depositing 0 in 7624 effectively disables the automatic pagination.
4.) Pages 1 & 3 of the appended listing were paginated automatically. Page 2 was ejected (CTRL/K + Return) after 51 lines, to make a cleaner break; page 4 was fattened with several CTRL/K + Return combinations.

Premature halts: If the user halts PAL-III in the middle of a pass, before it has reached the \$ sign at the end of the user's program, he should deposit 7600 in location 1403 before restarting. An alternative would be to reload the binary tape of Mate.

[Footnote: \$ signs in comments are legal and will not terminate assembly.]

Further notes: The elimination of paper tape input, for users of the low-speed (Teletype) reader, makes certain options practical for the first time.

- 1.) One can run a number of first passes through Pal while his source program is still incomplete. The listing of unidentified symbols is a help in writing the program.
- 2.) A second pass will quickly pick up illegal (off-page) references.
- 3.) Programs can be easily relocated if addresses are given symbolic names. It costs little time to assemble several versions of a routine, each with a different core location.
- 4.) Two third-pass listings can be printed in the time formerly required by one.
- 5.) General practice at the writer's installation is to punch a symbolic source tape after the program has been run, debugged, and listed. It saves time.

Timing: Something like 2 seconds is required to read a core page program from Editor's buffer.

/PROGRAM TO MATE PAL-III (FIELD 0 OR 1) WITH SYMBOL
/EDITOR (FIELD 1 OR 0). DOES NOT MODIFY EDITOR.

/MODIFICATIONS TO PAL-III ASSEMBLER:

/SUGGESTED CHANGE FOR PDP-8/E USERS ONLY:

/*224

/ 6040 (SPF) [TO INITIALIZE PRINTER]

*1401

1401 5603 JMP I .+2 /REPLACES 6031 (KSF)

*1403

1403 7600 BEGN /REPLACES 6036 (KRB)

/MAIN PROGRAM: RESIDES IN OCTAL PG. 37 OF PAL'S FIELD

*7600

7600	7600	BEGN,	.	
7601	1224		TAD M67	/55-LINE-PER-PAGE COUNT
7602	3225		DCA LINECT	
7603	4363		EDFLD	/CHANGE DATA FIELD
7604	1626		TAD I PTR0	/ADDR. OF 1ST PTR. IN EDITOR
7605	3375		MQL	
7606	4363		PALFLD	/CHANGE BACK TO CUR. FLD.
7607	1375		MQA	
7610	7410		SKP	
7611	1231	LINE,	TAD NEWPTR	
7612	3230		DCA PTR	
7613	3267		DCA MODE	/'MODE'=0 UNTIL SLASH IS READ
7614	4303		JMS FETCH	/GET WORD (ADRS. OF NEXT PTR.)
7615	3231		DCA NEWPTR	
7616	4273	GO,	JMS BYTE1	/BYTE #1 OF TEXT WORD
7617	4232		JMS ROUTIN	/CHECK, AUGMENT, DELIVER TO PAL
7620	1375		MQA	
7621	0227		AND K77	/BYTE #2
7622	4232		JMS ROUTIN	
7623	5216		JMP GO	/LOOP UNTIL (EXECUTABLE) "\$"
7624	7711	M67,	-67	/-55 DECIMAL; CHANGE AS DESIRED.
7625	0000	LINECT,	0	
7626	0112	PTR0,	112	/IN SYMBOL EDITOR
7627	0077	K77,	77	
7630	0000	PTR,	0	
7631	0000	NEWPTR,	0	/HOLDS POINTER TO NEXT LINE
7632	0000	ROUTIN,	0	
7633	1261		TAD M77	/FLAG FOR SPECIAL CHARS.
7634	7450		SNA	
7635	5312		JMP SVNTY7	
7636	1262		TAD K20	/SLASH?
7637	7440		SZA	
7640	5243		JMP NRML1	/NO

"MATE"--further notes

1. Versions of Pal and Editor

"Mate" was written for, and works with, paper tapes labeled as follows: Pal, "DEC-08-ASCI-PB 4/13/70" ; and Editor, "DEC-08-ESAC-PB 2/4/70" .

"Pal-III, 1972" appears to be the same as the 1970 Pal, except for an expanded symbol table which includes 8/E machine codes. It should therefore work with MATE.

Earlier versions than 1970 of the two programs probably will not work with MATE. The user can find where the KSF...KRB are in Pal by using ODT. The text pointer in Editor is location 112 in the version available to me; it is said to be 111 in earlier versions, but this I cannot verify. The number 112 is stored in loc. 7626 of MATE.

2. MATE in a high-speed reader installation

If you load core with the high-speed reader, the first attempt to start Pal-III (overlaid with MATE) will result in a halt. Load address = 0200 once more and start; this time it will work. Apparently the H.S.R. flag catches Pal's attention, so that it does not check the low-speed reader instructions (which are the ones overlaid by MATE).

Programs written for Pal-III must end in a \$ sign, whether one uses MATE or paper tape input. Failure to do this results in a bushel of illegal character diagnostics.

/-----
/ROUTINE FOR SPECIAL CHARACTERS:

7712	1232	SVNTY7,	TAD ROUTIN	
7713	7041		CIA	
7714	1362		TAD CALL1	/WHERE WAS 'ROUTIN' CALLED FROM?
		/('CALL1' IS RETURN ADDRESS FROM 1ST POSSIBLE CALL)		
7715	7640		SZA CLA	
7716	5353		JMP SECOND	
7717	1375	FIRST,	MQA	/77 WAS FIRST BYTE OF WORD
7720	0227		AND K77	
7721	4324		JMS CRTEST	/CODES (2)15 OR (2)13?
7722	4660		JMS I PAL	
7723	5216		JMP GO	
7724	0000	CRTEST,	0	
7725	1352		TAD M15	/CARR. RETURN?
7726	7440		SZA	
7727	5342		JMP NOTCR	/NO
		/.....		
7730	2225		ISZ LINECT	/55 LINES YET?
7731	5337		JMP .+6	/NO
7732	7001	PAGE,	IAC	/YES, SEND FORM FEED
7733	1350		TAD K213	
7734	4660		JMS I PAL	
7735	1224		TAD M67	/RESET LINE CTR.
7736	3225		DCA LINECT	
7737	1351		TAD K215	/RESTORE CR AND PASS
7740	4660		JMS I PAL	/ ON TO PAL-III
7741	5211		JMP LINE	/NEW LINE OF TEXT
		/.....		
7742	7001	NOTCR,	IAC	
7743	7001		IAC	/CHECK FOR CTRL/K
7744	7450		SNA	
7745	5332		JMP PAGE	/TO SIMULATE FORM CODE
7746	1350		TAD K213	
7747	5724		JMP I CRTEST	
7750	0213	K213,	213	
7751	0215	K215,	215	
7752	7763	M15,	-15	
7753	4273	SECOND,	JMS BYTE1	/READ NEW WORD, 1ST BYTE
7754	4324		JMS CRTEST	
7755	3273		DCA SAVE	
7756	1362		TAD CALL1	/AS IF FROM 1ST CALL OF "ROUTIN"
7757	3232		DCA ROUTIN	/RETURN ADDRS.
7760	1273		TAD SAVE	
7761	5256		JMP PAL-2	
7762	7620	CALL1,	GO+2	
		SAVE=BYTE1		
		/-----		

7641	7040	SLASH,	CMA	
7642	3267		DCA MODE	/TO INDICATE COMMENT MODE
7643	1263	NRML1,	TAD K13	/DOLLAR SIGN?
7644	7440		SZA	
7645	5252		JMP NRML2	/NO
7646	2267	DOLLAR,	ISZ MODE	/COMMENT MODE, OR PROGRAM?
7647	5270		JMP FINIS	/PROGRAM, \$ TERMINATES
7650	7040		CMA	/COMMENT; RESET 'MODE'
7651	3267		DCA MODE	/ TO -1 & PROCEED
7652	1264	NRML2,	TAD K4	/(AC=FRAME-40 NET)
7653	7510		SPA	/CODES<40 GET 300, OTHERS 200.
7654	1265		TAD K100	
7655	1266		TAD K240	/RESTORE 40 PLUS 200
7656	4660		JMS I .+2	/DELIVER TO PAL
7657	5632		JMP I ROUTIN	/PAL RETURNS HERE
7660	1403	PAL,	1403	
7661	7701	M77,	-77	
7662	0020	K20,	20	
7663	0013	K13,	13	
7664	0004	K4,	4	
7665	0100	K100,	100	
7666	0240	K240,	240	
7667	0000	MODE,	0	
7670	1200	FINIS,	TAD BEGN	
7671	3232		DCA ROUTIN	/THIS REROUTES PAL TO 7600
7672	5252		JMP NRML2	
7673	0000	BYTE1,	0	
7674	2230		ISZ PTR	/(SHD. NEVER REACH 7777)
7675	4303		JMS FETCH	
7676	7112		CLL RTR	
7677	7012		RTR	
7700	7012		RTR	
7701	0227		AND K77	
7702	5673		JMP I BYTE1	
7703	0000	FETCH,	0	
7704	4363		EDFLD	
7705	1630		TAD I PTR	
7706	3375		MQL	
7707	4363		PALFLD	
7710	1375		MQA	
7711	5703		JMP I FETCH	

BEGN	7600
BYTE1	7673
CALL1	7762
CRTEST	7724
DOLLAR	7646
EDFLD	4363
FETCH	7703
FINIS	7670
FIRST	7717
GO	7616
KCDF	7773
K10	7774
K100	7665
K13	7663
K20	7662
K213	7750
K215	7751
K240	7666
K4	7664
K77	7627
LINE	7611
LINECT	7625
MODE	7667
MQA	1375
SQL	3375
M15	7752
M67	7624
M77	7661
NEWPTR	7631
NOTCR	7742
NRML1	7643
NRML2	7652
PAGE	7732
PAL	7660
PALFLD	4363
PTR	7630
PTR0	7626
ROUTIN	7632
SAVE	7673
SECOND	7753
SLASH	7641
SVNTY7	7712

EDFLD= JMS .

PALFLD= JMS .

/SUBROUTINE TO SWITCH D.F. 0 & 1 (EITHER WAY):

7763	0000	0
7764	6214	RDF
7765	1374	TAD K10
7766	0374	AND K10
7767	1373	TAD KCDF
7770	3371	DCA .+1
7771	0000	0
7772	5763	JMP I .-7

7773	6201	KCDF,	CDF
7774	0010	K10,	10

MQL=DCA .

/SIMULATED 8/E CODES

MQA=TAD .

7775	0000	0
------	------	---